

# Chapter 24

## Matrix Programs

## 24.1 Matrix calculations

Examples of various matrix operations using SAS proc `iml` (SAS Institute Inc. 2018).

---

```
* matrix2.sas;
title 'Matrix calculations';
proc iml;
reset print;
* Define matrix A and B;
A = {1, 2, 3};
B = {4, 5, 6};
* Add A and B;
AplusB = A + B;
* Define matrix C and D;
C = {1 4, 2 5, 3 6};
D = {7, 8};
* Multiply C and D;
CD = C*D;
* Transpose of F;
F = {1 5, 2 6, 3 7, 4 8};
transposeF = t(F);
* Define another matrix A;
A = {1 6 4, 3 7 6, 4 1 9};
* Inverse of A;
Ainv = inv(A);
* Check that Ainv*A = I;
AinvA = Ainv*A;
quit;
```

---

## 24.2 Multiple regression in matrix form

A multiple regression analysis using matrix operations and `proc iml` (SAS Institute Inc. 2018). The data are from Reeve et al. (1998).

---

```
* multreg.sas;
title 'Multiple regression in matrix form';
data multdata;
    input X1 X2 survival;
    * Apply transformations here;
    Y = log(survival);
    datalines;
1.250 0.000 0.107
2.656 0.481 0.715
7.334 0.171 0.036
1.603 0.352 0.188
2.622 0.016 0.438
1.000 0.000 0.585
4.342 0.185 0.115
5.233 0.018 0.257
2.500 0.410 0.032
3.250 0.015 0.350
6.000 0.007 0.161
4.750 0.000 0.073
2.500 0.095 0.219
8.750 0.033 0.028
6.000 0.015 0.294
5.000 0.105 0.207
7.149 0.025 0.227
6.750 0.015 0.040
7.500 0.043 0.089
2.500 0.073 0.176
5.000 0.055 0.084
2.250 0.023 0.203
1.250 0.123 0.074
4.750 0.035 0.126
4.500 0.212 0.290
9.557 0.166 0.010
5.000 0.338 0.207
;
run;
* Print the data;
proc print data=multdata;
run;
```

```

* Matrix calculations;
proc iml;
reset print;
* Read in data set;
use multdata var {Y X1 X2};
read all;
close multdata;
* Design matrix X;
n = nrow(Y); * Find sample size;
ones = shape(1,n,1);
X = ones||X1||X2;
* Y values;
print Y;
* X' or X transpose;
Xt = t(X);
* X'X;
XtX = Xt*X;
* X'X inverse;
XtXinv = inv(XtX);
* Show this is the inverse;
test = XtXinv*XtX;
* (X'X inverse)X';
XtXinvXt = XtXinv*Xt;
* beta = (X'X inverse)X'Y;
beta = XtXinvXt*Y;
* Yhat;
Yhat = X*beta;
* SSerror and MSerror;
SSerror = sum((Y-Yhat)##2);
dfnum = nrow(beta)-1;
dfdenom = n - dfnum - 1;
MSerror = SSerror/dfdenom;
* SSreg and MSreg;
Ymean = mean(Y);
SSreg = sum((Yhat-Ymean)##2);
MSreg = SSreg/dfnum;
* F statistic and P value for overall test;
F = MSreg/MSerror;
P = 1 - probf(F,dfnum,dfdenom);
* Standard errors for beta;
sebeta = sqrt(MSerror*vecdiag(XtXinv));
* t tests for beta_i = 0;
Tvec = beta/sebeta;
Pvec = 2*(1-probt(abs(Tvec),dfdenom));
quit;

```

---

## 24.3 References

- Reeve, J. D., Rhodes, D. J. & Turchin, P. (1998) Scramble competition in southern pine beetle (Coleoptera: Scolytidae). *Ecological Entomology* 23: 433-443.
- SAS Institute Inc. (2018) *SAS/IML 15.1 User's Guide*. SAS Institute Inc., Cary, NC.

